      *

     FQMISRC  IF  F      92            DISK

      *

      *

     E                    SA       9999 80

      *

      *

     IQMISRC  NS

     I                                        1  12 SEQDAT

     I                                       13  92 SRCDTA

      *

      *

     IPOPTDS      DS

     I                                    B   1   20P#OPTS

     I                                        3 167 POPTS

      *

     IOPTDS       DS

     I                                        1  11 REPLAC

     I                                       12 176 OOPTS

      *

     IFIELDS     IDS

     I                                    B   1   40SRCLEN

     I I            1                     B   5   80PAGNBR

     I                                    B   9  120#OPTS

      *

     C           *ENTRY    PLIST

     C                     PARM           QPGM   20

     C                     PARM           PGMTXT 50

     C                     PARM           QSRCF  20

     C                     PARM           SRCMBR 10

     C                     PARM           SCHGDT 13

     C                     PARM           QPRTF  20

     C                     PARM           PUBAUT 10

     C                     PARM           POPTDS

     C                     PARM           PREPLC  1

      *

      *

      *  Read through the source file putting the source

      *  records in the source array SA, which will be

      *  passed to the API.

     C                     MOVEL*ZEROS    I       50

     C                     READ QMISRC                   LR

     C           *INLR     DOWEQ'0'

     C                     ADD  1         I

     C                     MOVELSRCDTA    SA,I

     C                     READ QMISRC                   LR

     C                     ENDDO

      *

      *

      *  Set source length.

     C           I         MULT 80        SRCLEN

      *

      *

      *  Set option template. Move options from incoming

      *  parm (POPTDS) to template parm passed to API

      *  (OPTDS).

     C           PREPLC    IFEQ '1'

     C                     MOVEL'*REPLACE'REPLAC

     C                     ELSE

     C           '*NO'     CAT  'REPLACE' REPLAC

     C                     ENDIF

      *

     C           P#OPTS    MULT 11        LEN     50

     C           LEN       SUBSTPOPTS:1   OOPTS

      *

     C           P#OPTS    ADD  1         #OPTS

      *

      *

      *  Call API.

     C                     CALL 'QPRCRTPG'

     C                     PARM           SA

     C                     PARM           SRCLEN

     C                     PARM           QPGM

     C                     PARM           PGMTXT

     C                     PARM           QSRCF

     C                     PARM           SRCMBR

     C                     PARM           SCHGDT

     C                     PARM           QPRTF

     C                     PARM           PAGNBR

     C                     PARM           PUBAUT

     C                     PARM           OPTDS

     C                     PARM           #OPTS

      *

      *

     C                     RETRN

      *




